# The Downward Continuation of Source and Receivers in Resistivity Surveying: Trying to Pose the Problem

by Vincent E. Courtillot

The problem we are interested in is the inverse problem of d.c. geoelectric sounding. We want to get as much information as we can on electric resistivity as a function of depth from surface data: these data are the value of injected current and the recorded values of potential at a number of locations. The way this problem is solved in general is by matching a set of master curves for the apparent resistivity of simple models to the observed apparent resistivity as a function of offset. We wanted to investigate an alternate solution through the use of the methods outlined in this report and the previous one.

The basic idea was to estimate local resistivity from the short offset data and then to use the principle of reciprocity in order to downward continue both the current (or potential) source and the receivers (which measure potential). The results presented in this note are still incomplete, but it is hoped that the remarks and discussions contained in it can help in future work.

I. We will start with a brief formulation of the forward problem in terms of matrizants for the simple case of layered media. The partial differential equations relating current density $\vec{J}$ , electric field $\vec{E}$ , conductivity $\sigma$ (which can be a scalar or a tensor), potential V and electric charge q are

$$\text{div } \vec{J} = - \partial q/\partial t \quad , \quad \vec{J} = \sigma \vec{E} \, , \quad \vec{E} = - \vec{\text{grad}}\, V \qquad (1)$$

Eliminate $\vec{E}$ , write equations for both horizontal and vertical components $J_x$ and $J_z$ of $\vec{J}$ , bring $\partial/\partial z$ terms on the left side and eliminate $J_x$ ; one finally gets:

$$\begin{cases} \dfrac{\partial V}{\partial z} = -\dfrac{1}{\sigma} J_z \\ \\ \dfrac{\partial J_z}{\partial z} = +\dfrac{\partial}{\partial x}\left( \sigma \dfrac{\partial V}{\partial x} \right) - \dfrac{\partial q}{\partial t} \end{cases} \tag{2}$$

We can assume (layered media hypothesis) that $\sigma$ does not depend on $x$ and Fourier transform (2) w.r. to $x$ (wave number $k_x$ ); the resulting equations can be put in matrix form:

$$\partial X / \partial z = A X + S \tag{3}$$

where the unknown vector $X$ is $( V, J_z )^t$ , the source term is $S = ( 0 , -\partial q / \partial t )^t$ and matrix $A$ is:

$$A = \begin{bmatrix} 0 & -1/\sigma \\ -k_x^2 \sigma & 0 \end{bmatrix} \tag{4}$$

This formulation is very similar to the one given by Claerbout for the wave propagation problem. One can use the matrix $( I + A \Delta z )$ to get $X$ from $z$ to $z + \Delta z$ and can repeat the operation in order to get from the top to the bottom (or conversely). The corresponding matrizant takes a very simple form in a source free layer, i.e., $\exp A[ z_{top} - z_{bottom}]$. This matrix is easily computed by using Sylvester's theorem on the diagonalized version of $A$ . The eigenvalues of $A$ are $|k_x|$ and $-| k_x|$ ; thus the solutions corresponding to up and downgoing waves in the case of the wave equation are $e^{|k_x|z}$ and $e^{-|k_x|z}$ ; these two solutions "propagate" separately in source free regions and only interact at interfaces ("reflectors"). It is easy to show that one can "propagate" X from the jth layer to the (j+1)st layer through:

$$\begin{pmatrix} V \\ J_z \end{pmatrix}_{j+1} = \begin{bmatrix} 1/\sigma_j & 1/\sigma_j \\ -|k_x| & |k_x| \end{bmatrix} \begin{bmatrix} e^{|k_x|h_j} & 0 \\ 0 & e^{-|k_x|h_j} \end{bmatrix} \begin{bmatrix} \sigma_j/2 & -1/2|k_x| \\ \sigma_j/2 & 1/2|k_x| \end{bmatrix} \begin{pmatrix} V \\ J_z \end{pmatrix}_j \quad (5)$$

One can also see that the vector behaving like $( D, U )^t$ , where D is the downgoing and U the upgoing wave, is

$\left( \frac{1}{2} (\sigma V - J_z / |k_x|) , \frac{1}{2} (\sigma V + J_z / |k_x|) \right)^t$ and that:

$$\begin{pmatrix} D \\ U \end{pmatrix}_2 = \frac{\sigma_1 + \sigma_2}{2\sigma_1} \begin{bmatrix} 1 & \frac{\sigma_2 - \sigma_1}{\sigma_2 + \sigma_1} \\ \frac{\sigma_2 - \sigma_1}{\sigma_2 + \sigma_1} & 1 \end{bmatrix} \begin{bmatrix} e^{|k_x|h_1} & 0 \\ 0 & e^{-|k_x|h_1} \end{bmatrix} \begin{pmatrix} D \\ U \end{pmatrix}_1 \quad (6)$$

This shows the correspondence between impedance in wave propagation problems (recall that the reflection coefficient is $\frac{I_1 - I_2}{I_1 + I_2}$) and resistivity in d.c. geoelectric soundings.

By repeating (5) and (6) one gets the classical formula for layered media. Such equations can be used to downward continue the receivers. We will see an alternate way of doing this in the last part of this note, where a resistor network approximation to the continuous case is described.

II. In this paragraph we will describe the general way we intend to perform the simultaneous downward continuation of sources and receivers. This will raise two major problems which will be discussed in the last paragraph.

Assume that we start with the following experiment on the surface where S is the source and R is the receiver

z = 0    R R R S R R R    Fig. 1

We can hope to estimate the resistivity of the earth in the immediate vicinity of S from the short offset data. For a continuous homogeneous half space the potential V is given as a function of horizontal distance r from S by $V = \rho I / 2\pi r$ ; we can expect to obtain a local average of $\rho$ from the limit of $2\pi r V / I$ as r goes to 0 . A first problem we will have to solve is to obtain such a local approximation of $\rho$ in the case of finite sampling interval (we can hope to find a solution to this problem by working on finite resistor network approximation of the earth).

Once resistivity has been estimated in the vicinity of S (that is in a layer of thickness $\Delta z$ – since we make a layered media hypothesis), the readings from the receivers can be downward continued, using equation (5). We now have the results from the following fictitious experiment:




Fig. 2

The next idea is to apply the principle of reciprocity for a (S,R) pair. The correct formulation of the reciprocity principle happens to be the second problem we are faced with; it will be discussed in the next paragraph.

Applying the principle for one pair means we can get the results of the following experiment:




Fig. 3

It can be repeated for all receivers of figure 2, and when correctly rearranged leads to:




Fig. 4

Now, equation (5) can be used to downward continue the receivers by $\Delta z$ when the source term for S is <u>correctly</u> included (this is another problem).

If we can perform all these steps, we now have an experiment similar to that of figure 1, except that both sources and receivers have been downward continued by $\Delta z$ . From this one can estimate the next value of resistivity and repeat the whole process as deep as desired (or as possible if there are stability problems).

In the next paragraph we will try to discuss the problems that arose in this paragraph, starting with a discussion on how to apply reciprocity.

III. In order to get a correct statement of how reciprocity can be applied, we shall study the simple case of a finite resistor network as illustrated below (all resistors on the branches have value r ):




Fig. 5

When all currents and voltage sources are specified we have 29 unknowns: the values of potential at the 12 nodes, 9 horizontal currents i and 8 vertical currents j . Upon expressing Kirchoff's laws ( $\Sigma$ intensities at one node = 0 ; $\Sigma \Delta V$ around one loop = $\Sigma$ e.m.f ) one gets the following matrix equation:

$$\left[\begin{array}{cc|c} \begin{matrix} -r & & 0 \\ & \ddots & \\ 0 & & -r \end{matrix} & & A \\ \hline A^t & & 0 \end{array}\right] \times \begin{bmatrix} i \\ \hline j \\ \hline V \end{bmatrix} = \begin{bmatrix} \varepsilon_x \\ \hline \varepsilon_z \\ \hline I \end{bmatrix} \begin{matrix} \}\, 9 \\ \}\, 8 \\ \}\, 12 \end{matrix} \qquad (7)$$

where block $A$ contains mostly zeros and a few 1 and $-1$ ; $\varepsilon_x$ are the voltage sources on horizontal segments, $\varepsilon_z$ the voltage sources on vertical segments and $I$ the current inputs and outputs at the various nodes; we require of course that the sum of the $I$'s be zero. (7) can be rewritten as:

$$\left.\begin{aligned} -r \begin{bmatrix} i \\ j \end{bmatrix} + AV &= \begin{bmatrix} \varepsilon_x \\ \varepsilon_z \end{bmatrix} \\ A^t \begin{bmatrix} i \\ j \end{bmatrix} &= I \end{aligned}\right\} \Longrightarrow \quad \begin{aligned} BV = A^t \begin{bmatrix} \varepsilon_x \\ \varepsilon_z \end{bmatrix} + rI = S \\ \text{with} \\ B = A^t A \end{aligned}$$

where $S$ is the source vector and <u>$B$ is a symmetric matrix</u>. But $B$ is not a full rank matrix, since one potential can be defined arbitrarily: we can replace the last line of $B$ by the equation $V_{12} = 0$ , for example.

In that case we can also drop the last column of B , since all its elements multiply $V_{12}$ and drop the last term in V and S . We now have the full rank system:

$$B'V' = S' \quad , \quad \text{together with} \quad V_{12} = 0 \quad . \qquad (9)$$

Program NETWOR, listed at the end of this note, is designed to create matrices B' and S' and solve system (9). The rectangular network consists of NN nodes on a horizontal segment and IMAX on a vertical segment. Thus, there are NV1 = NN * IMAX - 1 unknown independent potentials V . The ordering of nodes is done according to the scheme of figure 5; the last node in the lower right hand corner is arbitrarily taken to have zero potential.*

The page following the program listing shows the solution of the example described in figure 5. Source currents and potentials are printed in a way that makes them readily understandable. The currents can also be computed and they show the antisymmetry that could be expected:




* Note: Subroutine GOLUB, written by Don Riley, is used to finally solve (9). The reader may be interested in studying this subroutine which allows one to compute the least squares solution of an overdetermined system by the method of Gene Golub.

Also of interest is the structure of matrices A and B (see page 171); this structure can quickly be checked in the simple example of figures 5 and 6.

Let us now see how the structure of system (9) helps in understanding how the principle of reciprocity should be applied; let us perform a first experiment with sources $S_1'$ , and response $V_1'$ , and a second experiment with $S_2'$ and $V_2'$ . It is easy to see that the symmetry of $B'$ implies:

$$S_1'^t \, V_2' = S_2'^t \, V_1' \qquad (10)$$

Each side of equation (10) contains three kinds of terms: terms such as $i \varepsilon_1$ , $j \varepsilon_2$ and $I V$ . In what follows we shall assume that there are no potential sources ( $\varepsilon_1 = \varepsilon_2 = 0$ ) .

In (10) , one node has been set aside, generating a certain asymmetry: we would like to get rid of this asymmetry in our formulation of the reciprocity principle. This can be done in the following fashion: inject $I$ at node 2 (for example) and remove it at node 3 , then measure $V_7$ and $V_8$ (for example); in a second experiment inject $I'$ at node 7 , remove it at node 8 and measure $V_2'$ and $V_3'$ . Then reciprocity implies that:

$$I \, ( V_2' - V_3' ) = I' \, ( V_7 - V_8 ) \qquad (11)$$

From (11), it becomes clear that only differences in potential have to be used; the reference point for $V$ can be arbitrarily chosen. We shall use (11) with $I=I'$ , in order to <u>downward continue the dipolar current source</u> as outlined in paragraph II. The algorithm is as follows:

- Read the surface data: NN values of potential and the intensity CUR of the injected current.
- Make a first local estimate of resistance R
- Use this estimate to downward continue V by one depth interval (over which the estimate of R is supposed valid). This can be done using simply Kirchoff's laws. We then have the following fictitious experiment:




Fig. 7

- Apply reciprocity to the pair of nodes where current is injected and extracted, and to all pairs of nodes at the Δz level (dashed contours). We then have the following fictitious experiment (the reference level for V is arbitrary):




Fig. 8

- Downward continue V by one Δz step.
- From this full downward continued experiment, get the next estimate of resistance R (short offset data).
- Use this new estimate of R to downward continue V by one more depth interval in the first frame (fig. 7):



- Apply reciprocity between levels 0 and $2\Delta z$ . The whole procedure can now be repeated: potentials are downward continued by $2\Delta z$ on the new frame, a new estimate of R at depth $2\Delta z$ is obtained and one can come back to the first frame and downward continue V by one more step, etc....

Program RHO , listed after NETWOR at the end of this note, performs all these operations. This program is still in a preliminary stage: the subroutine which should estimate local values of R (derived from the continuous formula given on page 174) has not yet been written. Synthetic data are obtained from NETWOR (in that case NN = 20 and IMAX = 6 ) and read by RHO , which then downward continues them. The estimating step is bypassed and replaced by a correct evaluation of resistance ( RR , known of course since this is a synthetic example).

Results of a test run are listed in tables 1 and 2; table 1 shows the potentials computed with NETWOR (only the left half is shown). The first line of table 1 was used as data in RHO and the output of RHO is shown in table 2. The first four lines, that is the first four depths, are in reasonably close agreement, but start diverging afterwards. This is due to the fact that the input data had an important roundoff error, which propagated downwards. Otherwise, one can check that Kirchoff's laws are accurately verified in table 2. However, this instability in the continuation is a major problem and must be studied in detail before further progress can be made.

Concluding remarks: Besides having an interest of its own (for the inverse resistivity problem is an important and interesting one indeed), it is hoped that the remarks contained in this note may help in understanding how the simultaneous downward continuation of both sources and receivers in the active medium, the unknown earth structure, can be implemented. This understanding hopefully can be extended to the problem of downward continuation of the sources in reflexion seismic exploration, which was discussed briefly in the March '74 report.




Resistor network (NN=20,IMAX=6,CUR=1.)

| | | | | | | | | | |
|---|---|---|---|---|---|---|---|---|---|
| 0.166 | 0.166 | 0.167 | 0.170 | 0.174 | 0.181 | 0.194 | 0.219 | 0.273 | 0.403 |
| 0.165 | 0.166 | 0.166 | 0.168 | 0.170 | 0.175 | 0.181 | 0.191 | 0.198 | 0.176 |
| 0.164 | 0.164 | 0.165 | 0.165 | 0.166 | 0.166 | 0.166 | 0.164 | 0.152 | 0.118 |
| 0.163 | 0.163 | 0.163 | 0.162 | 0.161 | 0.159 | 0.154 | 0.145 | 0.129 | 0.101 |
| 0.162 | 0.162 | 0.161 | 0.160 | 0.157 | 0.153 | 0.146 | 0.135 | 0.118 | 0.094 |
| 0.162 | 0.161 | 0.160 | 0.158 | 0.155 | 0.150 | 0.142 | 0.130 | 0.114 | 0.092 |

TABLE 1

| | | | | | | | | | |
|---|---|---|---|---|---|---|---|---|---|
| 0.166 | 0.166 | 0.167 | 0.170 | 0.174 | 0.181 | 0.194 | 0.219 | 0.273 | 0.403 |
| 0.165 | 0.165 | 0.166 | 0.168 | 0.171 | 0.174 | 0.181 | 0.191 | 0.198 | 0.176 |
| 0.164 | 0.164 | 0.165 | 0.165 | 0.167 | 0.165 | 0.166 | 0.164 | 0.152 | 0.103 |
| 0.164 | 0.162 | 0.164 | 0.159 | 0.165 | 0.155 | 0.155 | 0.147 | 0.143 | 0.085 |
| 0.165 | 0.156 | 0.172 | 0.142 | 0.181 | 0.133 | 0.152 | 0.124 | 0.189 | 0.094 |
| 0.176 | 0.124 | 0.224 | 0.058 | 0.283 | 0.043 | 0.197 | 0.008 | 0.395 | 0.101 |

TABLE 2

PROGRAM NETWOR

```
      DIMENSION A(17,11),B(11,11),S(11)
      DIMENSION AT(11,17)
      DIMENSION V(11),U(11)
C*************************************************
C  THIS PROGRAM COMPUTES THE POTENTIALS V AT THE
C  NV NODES OF A NN BY IMAX RESISTOR NETWORK.
C  NN IS THE NUMBER OF SURFACE POINTS,IMAX THE
C  NUMBER OF DEPTH LEVELS. POTENTIAL IS ARBITRA-
C  RILY SET TO ZERO IN THE LOWER RIGHT HAND
C  CORNER. ALL RESISTORS ARE EQUAL TO R. CURRENT
C  CUR IS INJECTED IN THE NETWORK ACCORDING TO
C  SOURCE DISTRIBUTION S.
C    STANFORD SEPT 74      V.E.COURTILLOT
C*************************************************
      NN=4
      IMAX=3
      NCH=(NN-1)*IMAX
      NCV=NN*(IMAX-1)
      NC=NCH+NCV
      NV=NN*IMAX
      NV1=NV-1
      R=1.
      CUR=1.
      WRITE(6,990)NN,IMAX,NC,NV1
  990 FORMAT(1X,'NUMBER OF SURFACE POINTS',I5,5X,'NUMBER OF DEPTH ',
     1'POINTS',I5,/,1X,'NUMBER OF UNKNOWN CURRENTS',I5,5X,'NUMBER OF ',
     2'UNKNOWN POTENTIALS',I5,/)
      WRITE(6,991)R,CUR
  991 FORMAT(1X,'VALUE OF RESISTANCE',F5.2,5X,'INTENSITY OF ',
     1'INJECTED CURRENT',F5.2,/)
C  CREATE MATRIX A
      DO 10 I=1,NC
      DO 10 J=1,NV1
   10 A(I,J)=0.
      KMAX=IMAX-1
      DO 100 K=1,KMAX
      K1=(K-1)*(NN-1)+1
      K2=K*(NN-1)
      DO 20 I=K1,K2
      A(I,I+K-1)=-1.
   20 A(I,I+K)=1.
  100 CONTINUE
      I1=(IMAX-1)*(NN-1)+1
      I2=IMAX*(NN-1)-1
      DO 21 I=I1,I2
      A(I,I+IMAX-1)=-1.
   21 A(I,I+IMAX)=1.
      A(NCH,NV1)=-1.
      NC1=NCH+1
      NC2=NC-1
      DO 200 I=NC1,NC2
      A(I,I-NCH)=-1.
  200 A(I,I-NCH+NN)=1.
      A(NC,NCV)=-1.
C  CREATE SOURCE VECTOR S
      DO 300 I=1,NV1
  300 S(I)=0.
      S(2)=R*CUR
      S(3)=-R*CUR
```

```
      WRITE(6,992)
  992 FORMAT(1X,'VECTOR OF SOURCE CURRENTS',/)
      WRITE(6,9920) S
 9920 FORMAT(1X,4F11.3)
C  COMPUTE B=A(TRANSPOSED)*A
      CALL MATPRO(A,B,NC,NV1,AT)
      WRITE(6,993)
  993 FORMAT(1X,'MATRIX A',/)
      DO 700 I=1,NC
  700 WRITE(6,9930)(A(I,J),J=1,NV1)
 9930 FORMAT(1X,11F5.2)
      WRITE(6,994)
  994 FORMAT(1X,'MATRIX A*A TR',/)
      WRITE(6,9940) B
 9940 FORMAT(1X,11F5.2)
C  SOLVE SYSTEM B*V=S
      CALL GOLUB(B,V,S,NV1,NV1,U)
      WRITE(6,995)
  995 FORMAT(1X,'SOLUTION FOR POTENTIALS',/)
      WRITE(6,9950) V
 9950 FORMAT(1X,4F11.3)
      STOP
      END
      SUBROUTINE MATPRO(A,B,M,N,AT)
      DIMENSION A(M,N),B(N,N),AT(N,M)
C  THIS SUBROUTINE PREMULTIPLIES MATRIX A BY ITS TRANSPOSED
C  THE RESULTING MATRIX IS B
      DO 5 I=1,M
      DO 5 J=1,N
    5 AT(J,I)=A(I,J)
      DO 10 IR=1,N
      DO 10 IC=1,N
      B(IR,IC)=0.0
      DO 10 I=1,M
   10 B(IR,IC)=B(IR,IC)+AT(IR,I)*A(I,IC)
      RETURN
      END
      SUBROUTINE GOLUB (A,X,B,M,N,U)
C.....
C.....A(M,N) ; B(M)  GIVEN WITH M>=N  SOLVES FOR X(N) SUCH THAT
C.....               || B - AX || = MINIMUM
C.....METHOD OF G.GOLUB, NUMERISCHE MATHEMATIK 7,206-216 (1965)
C.....
      IMPLICIT REAL*8 (D)
      REAL*4 A(M,N),X(N),B(M),U(M)
C.....
C.....  PERFORM N ORTHOGONAL TRANSFORMATIONS TO A(.,.) TO
C.....  UPPER TRIANGULARIZE THE MATRIX
C.....
C.....
C.....
      NMAX=N
      IF(N.EQ.M) NMAX=N-1
      DO 3010 K=1,NMAX
C.....
      DSUM=0.0D0
      DO 1010 I=K,M
      DAJ=A(I,K)
 1010 DSUM=DSUM+DAJ**2
      IF(DSUM.GT.1.0D-10) GO TO 1015
```

```
      WRITE(6,900) DSUM,K
  900 FORMAT('0SINGULARITY IN GOLUB DSUM=',D15.6,5X,I4)
      GO TO 3010
 1015 CONTINUE
      DAI=A(K,K)
      DSIGMA=DSIGN(DSQRT(DSUM),DAI)
      DBI=DSQRT(1.0D0+DAI/DSIGMA)
      DFACT=1.0D0/(DSIGMA*DBI)
      U(K)=DBI
      FACT=DFACT
      KPLUS=K+1
      DO 1020 I=KPLUS,M
 1020 U(I)=FACT*A(I,K)
C.....
C.....  I - UU' IS A SYMMETRIC, ORTHOGONAL MATRIX WHICH WHEN APPLIED
C.....  TO A(.,.) WILL ANNIHILATE THE ELEMENTS BELOW THE PIVOT
C.....  DIAGONAL K
C.....
      DO 2030 J=K,N
C.....
C.....  APPLY THE ORTHOGONAL TRANSFORMATION
C.....
      FACT=0.0
      DO 2010 I=K,M
 2010 FACT=FACT+U(I)*A(I,J)
C.....
      DO 2020 I=K,M
 2020 A(I,J)=A(I,J)-FACT*U(I)
 2030 CONTINUE
C.....
      FACT=0.0
      DO 2040 I=K,M
 2040 FACT=FACT+U(I)*B(I)
C.....
      DO 2050 I=K,M
 2050 B(I)=B(I)-FACT*U(I)
C.....
 3010 CONTINUE
C.....
C.....  BACK SUBSTITUTE TO RECURSIVELY YIELD X(.)
C.....
      X(N)=B(N)/A(N,N)
      LIM=N-1
C.....
      DO 4020 I=1,LIM
      IROW=N-I
      SUM=0.0
C.....
      DO 4010 J=1,I
 4010 SUM=SUM+X(N-J+1)*A(IROW,N-J+1)
C.....
 4020 X(IROW)=(B(IROW)-SUM)/A(IROW,IROW)
C.....
      RETURN
      END
```

<u>**Results from a run of NETWOR**</u> **(source configuration of figure 5)**

```
COMPILE TIME =      0.51 SECONDS, OBJECT CODE=    5,896 BYTES
NUMBER OF SURFACE POINTS     4      NUMBER OF DEPTH POINTS    3
NUMBER OF UNKNOWN CURRENTS   17     NUMBER OF UNKNOWN POTENTIALS    11

VALUE OF RESISTANCE 1.00     INTENSITY OF INJECTED CURRENT 1.00

VECTOR OF SOURCE CURRENTS

       0.000        1.000       -1.000        0.000
       0.000        0.000        0.000        0.000
       0.000        0.000        0.000
MATRIX A

-1.00 1.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00
 0.00-1.00 1.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00
 0.00 0.00-1.00 1.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00
 0.00 0.00 0.00 0.00-1.00 1.00 0.00 0.00 0.00 0.00 0.00
 0.00 0.00 0.00 0.00 0.00-1.00 1.00 0.00 0.00 0.00 0.00
 0.00 0.00 0.00 0.00 0.00 0.00-1.00 1.00 0.00 0.00 0.00
 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00-1.00 1.00 0.00
 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00-1.00 1.00
 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00 0.00-1.00
-1.00 0.00 0.00 0.00 1.00 0.00 0.00 0.00 0.00 0.00 0.00
 0.00-1.00 0.00 0.00 0.00 1.00 0.00 0.00 0.00 0.00 0.00
 0.00 0.00-1.00 0.00 0.00 0.00 1.00 0.00 0.00 0.00 0.00
 0.00 0.00 0.00-1.00 0.00 0.00 0.00 1.00 0.00 0.00 0.00
 0.00 0.00 0.00 0.00-1.00 0.00 0.00 0.00 1.00 0.00 0.00
 0.00 0.00 0.00 0.00 0.00-1.00 0.00 0.00 0.00 1.00 0.00
 0.00 0.00 0.00 0.00 0.00 0.00-1.00 0.00 0.00 0.00 1.00
 0.00 0.00 0.00 0.00 0.00 0.00 0.00-1.00 0.00 0.00 0.00
MATRIX A*A TR

 2.00-1.00 0.00 0.00-1.00 0.00 0.00 0.00 0.00 0.00 0.00
-1.00 3.00-1.00 0.00 0.00-1.00 0.00 0.00 0.00 0.00 0.00
 0.00-1.00 3.00-1.00 0.00 0.00-1.00 0.00 0.00 0.00 0.00
 0.00 0.00-1.00 2.00 0.00 0.00 0.00-1.00 0.00 0.00 0.00
-1.00 0.00 0.00 0.00 3.00-1.00 0.00 0.00-1.00 0.00 0.00
 0.00-1.00 0.00 0.00-1.00 4.00-1.00 0.00 0.00-1.00 0.00
 0.00 0.00-1.00 0.00 0.00-1.00 4.00-1.00 0.00 0.00-1.00
 0.00 0.00 0.00-1.00 0.00 0.00-1.00 3.00 0.00 0.00 0.00
 0.00 0.00 0.00 0.00-1.00 0.00 0.00 0.00 2.00-1.00 0.00
 0.00 0.00 0.00 0.00 0.00-1.00 0.00 0.00-1.00 3.00-1.00
 0.00 0.00 0.00 0.00 0.00 0.00-1.00 0.00 0.00-1.00 3.00
SOLUTION FOR POTENTIALS

       0.348        0.441       -0.236       -0.143
       0.255        0.211       -0.006       -0.050
       0.205        0.155        0.050

EXECUTION TIME =      1.28 SECONDS
```

PROGRAM RHO

```
      DIMENSION V(10,20),R(10),XCUR(9,19),ZCUR(9,20),W(20),VV(20)
C*****************************************************************
C  THIS PROGRAM ESTIMATES RESISTANCE R AS A FUNCTION OF DEPTH
C  IDEP DOWN TO A MAXIMUM DEPTH IMAX WHEN A CURRENT CUR IS
C  INJECTED AT THE SURFACE AND POTENTIALS V(1,J) ARE RECORDED
C  AT NN SURFACE POINTS. XCUR(I,J) IS THE HORIZONTAL CURRENT AT
C  DEPTH I BETWEEN POINTS J AND J+1. ZCUR(I,J) IS THE VERTICAL
C  CURRENT AT DEPTH I AND POINT J. V(I,J) IS THE POTENTIAL AT
C  DEPTH I FOR POINT J.
C*****************************************************************
      CUR=1.
      N=10
      N1=N-1
      N2=N+2
      N3=2*N-1
      NN=2*N
      IMAX=6
      RR=1.
C     READ V(1,J) THEN ESTIMATE R(1)
      READ(5,800) VV
  800 FORMAT(10F6.4)
      WRITE(6,801) VV
  801 FORMAT(1X,10F8.4)
      DO 80 J=1,NN
   80 V(1,J)=VV(J)
      R(1)=RR
C  COMPUTE FIRST HORIZONTAL CURRENTS
      DO 10 J=1,N3
   10 XCUR(1,J)=(V(1,J+1)-V(1,J))/R(1)
C  VERTICAL CURRENTS AT ENDS
      ZCUR(1,1)=-XCUR(1,1)
      ZCUR(1,NN)=XCUR(1,N3)
C  VERTICAL CURRENTS AT SOURCE POINTS
      ZCUR(1,N)=XCUR(1,N-1)-XCUR(1,N)-CUR
      ZCUR(1,N+1)=XCUR(1,N)-XCUR(1,N+1)+CUR
C  VERTICAL CURRENTS (GENERAL)
      DO 20 J=2,N1
   20 ZCUR(1,J)=XCUR(1,J-1)-XCUR(1,J)
      DO 30 J=N2,N3
   30 ZCUR(1,J)=XCUR(1,J-1)-XCUR(1,J)
C  NEW POTENTIALS ONE Z STEP DOWN
      DO 40 J=1,NN
      V(2,J)=V(1,J)+R(1)*ZCUR(1,J)
   40 VV(J)=V(2,J)
      WRITE(6,801) VV
C  APPLY RECIPROCITY FOR ONE Z STEP
      W(1)=0.
      DO 50 J=1,N3
   50 W(J+1)=W(J)+V(2,NN-J+1)-V(2,NN-J)
C  DOWNWARD CONTINUE W BY ONE Z STEP ON NEW FRAME
      CALL DOWN(1,NN,N3,R,W)
C  ESTIMATE NEW VALUE OF RESISTANCE FROM W
C     CALL ESTIM(CUR,W,RR,N,...)
      R(2)=RR
      IF(IMAX.LE.1)GO TO 900
      JMAX=IMAX-1
C  DOWNWARD CONTINUATION OF V ON THE ORIGINAL FRAME
```

```
      DO 100 IDEP=2,JMAX
C  NEW HORIZONTAL CURRENTS
      DO 110 J=1,N3
  110 XCUR(IDEP,J)=(V(IDEP,J+1)-V(IDEP,J))/R(IDEP)
C  NEW VERTICAL CURRENTS
      ZCUR(IDEP,1)=-XCUR(IDEP,1)+ZCUR(IDEP-1,1)
      ZCUR(IDEP,NN)=XCUR(IDEP,N3)+ZCUR(IDEP-1,NN)
      DO 120 J=2,N3
  120 ZCUR(IDEP,J)=XCUR(IDEP,J-1)-XCUR(IDEP,J)+ZCUR(IDEP-1,J)
C  NEW POTENTIAL
      DO 140 J=1,NN
      V(IDEP+1,J)=V(IDEP,J)+R(IDEP)*ZCUR(IDEP,J)
  140 VV(J)=V(IDEP+1,J)
      WRITE(6,801) VV
C  APPLY RECIPROCITY FOR IDEP Z STEPS
      W(1)=0.
      DO 150 J=1,N3
  150 W(J+1)=W(J)+V(IDEP+1,NN-J+1)-V(IDEP+1,NN-J)
C  DOWNWARD CONTINUE W BY IDEP Z STEPS ON NEW FRAME
      CALL DOWN(IDEP,NN,N3,R,W)
C  ESTIMATE NEW VALUE OF RESISTANCE FROM W
C     CALL ESTIM(CUR,W,RR,N,...)
      R(IDEP+1)=RR
  100 CONTINUE
  900 STOP
      END
      SUBROUTINE DOWN(IDEP,NN,N3,R,W)
C**************************************************************
C  THIS SUBROUTINE DOWNWARD CONTINUES A POTENTIAL W GIVEN AT
C  NN POINTS ON THE SURFACE TO DEPTH IDEP. THE SOURCE IS AT
C  DEPTH IDEP AND THE NEW POTENTIAL THERE IS PUT BACK IN W.
C**************************************************************
      DIMENSION V(10,20),R(10),XCUR(9,19),ZCUR(9,20),W(20)
      DO 10 J=1,N3
   10 XCUR(1,J)=(W(J+1)-W(J))/R(1)
      ZCUR(1,1)=-XCUR(1,1)
      ZCUR(1,NN)=XCUR(1,N3)
      DO 20 J=2,N3
   20 ZCUR(1,J)=XCUR(1,J-1)-XCUR(1,J)
      DO 30 J=1,NN
   30 V(2,J)=W(J)   +ZCUR(1,J)*R(1)
      IF(IDEP.LE.1) GO TO 900
      JDEP=IDEP-1
      DO 100 IND=2,JDEP
      DO 110 J=1,N3
  110 XCUR(IND,J)=(V(IND,J+1)-V(IND,J))/R(IND)
      ZCUR(IND,1)=-XCUR(IND,1)+ZCUR(IND-1,1)
      ZCUR(IND,NN)=XCUR(IND,N3)+ZCUR(IND-1,NN)
      DO 120 J=2,N3
  120 ZCUR(IND,J)=XCUR(IND,J-1)-XCUR(IND,J)+ZCUR(IND-1,J)
      DO 130 J=1,NN
  130 V(IND+1,J)=V(IND,J)+R(IND)*ZCUR(IND,J)
  100 CONTINUE
      DO 700 J=1,NN
  700 W(J)=V(IDEP,J)
  900 RETURN
      END
```